
Abstract—Code obfuscation involves transforming a program
to a new version that performs the same computation but
hides the functionality of the original code. An important
property of such a transformation is that it preserves the
behavior of the original program. In this paper, we lay the
foundation for studying and reasoning about code obfus-
cating transformations, and show how the preservation of
certain behaviours may be formally verified. To this end,
we apply techniques of formal specification and verification
using the Coq Proof Assistant. We use and extend an existing
encoding of a simple imperative language in Coq along with
an encoding of Hoare logic for reasoning about this language.
We formulate what it means for a program’s semantics to be
preserved by an obfuscating transformation, and give formal
machine-checked proofs that these behaviours or properties
hold. We also define a lower-level block-structured language
which is “wrapped around” our imperative language, allow-
ing us to model certain flattening transformations and treat
blocks of codes as objects in their own right.

Index Terms—obfuscation, verification, security, correctness,
Coq, proof

1. Introduction

We expect our software systems to function correctly.
By “correctly”, we mean that a system will behave accord-
ing to explicit and/or implicit expectations, i.e., its writ-
ten and/or unwritten specifications. Typically, extensive
testing is done to increase the confidence in the correct
functionality of a piece of software.

The more testing that is done, the more confidence
one has of the likelihood of correctness, but positive
testing results are not a proof of correctness. In systems
where more assurance of correctness is required, various
types of deductive reasoning are often used, where for-
mal verification methods based on theoretical foundations
rooted in logic are employed. This is the approach that we
adopt here. It is important to note that formal verification
transfers the problem of confidence in program correctness
to the problem of confidence in specification correctness.
Thus, it is not a silver bullet; however, since specifications
are often smaller and less complex to express, we are able
to successfully reduce the trusted computing base (TCB)
and increase our chances of achieving correctness.

A common approach to formal verification used to
show a (software) system behaves according to its spec-
ification is to employ a specification language based on

the familiar use of assertions [8], [10]. A specification (in
the form of statements about program fragments that are
expected to hold) is typically expressed in some variation
of first-order logic, and the verification system will try to
either prove the assertions correct or signal that they don’t
hold. This can be a rather elaborate process. For example,
assertions can be used to generate logical formulas called
verification conditions (VCs), which are either fed into a
satisfiability modulo theories (SMT) [9] solver behind the
scenes in a verification backend, or are presented in a more
visible manner to a verification expert who will manually
discharge them. VC generation for program verification
goes back to at least Hoare logic [10], Eiffel style contracts
[15], and proof-carrying-code (PCC) [17].

In this paper, we use simple program fragments (not
considered real-world but which are the basis of real-world
programs) and a formal verification approach to the task
of program transformations. In particular, the question
we address here is: how can one perform obfuscating
or optimizing transformations on code that is assumed
to be correct with respect to certain specified behaviour
(expressed in some assertion language) while preserving
the correctness of the specified behaviour?

To achieve the highest level of assurance that a trans-
formation has maintained correctness of the code, one
should prove that the two versions of the program (before
and after the transformation) are equivalent. Although
there exist large-scale verification results such as the for-
mally verified compiler CompCert [11], in general scaling
up verification efforts to realistic systems is extremely
hard. CompCert involved the verification of a semantic
equivalence between C and a generated assembly lan-
guage, and took several man-years to complete.

Equivalence between two versions of a program cer-
tainly implies the correctness of any program properties
of interest proved for one version applies to the other.
Alternatively, what if we limit ourselves to only proving
properties of interest in the “before” version of a program,
and show that these properties are maintained in the
“after” version of the program after a particular transfor-
mation is applied? Here, our focus is on certain simple
transformations that don’t invalidate properties about the
“before” version versus the “after” version. For example,
consider the program snippet in Listing 1, which asserts
that (y > 2) and is easily verified visually. The snippet
in Listing 2 illustrates a simple obfuscating transformation
called variable splitting, where we have split the variable
x into two variables x1 and x2, and it is clear (visually)



that the assertion (y > 2) still holds.

x = 2; y = 5;

y = x + y;

assert(y > 2);

Listing 1: Original Code

x1 = 1; x2 = 1; y = 5;

y = x1 + y; y = x2 + y;

assert(y > 2);

Listing 2: Obfuscated Code

In general, though, most transformations, whether op-
timizations or obfuscations, invalidate assertions that hold
true about the “before” version. Obfuscation is particularly
troublesome, because its main goal is to hide the func-
tionality of the code from prying eyes while maintaining
the functionality of the “before” program. More formally,
according to Barak et al. as cited in [1], obfuscators are
programs that transform an input program (e.g. similar to
a compiler) into an output program such that the output
program satisfies the following three properties:

1) it is semantically equivalent to the input program
(functionality property);

2) it is at most polynomially bigger or slower than the
input program (slowdown property); and

3) it is as “hard to analyze and de-obfuscate” as a
blackbox version of the program (virtual black-box
property).

“Prying eyes” could, for example, be some kind of static
analysis tool where an attacker is trying to determine
certain facts about the code, and obfuscation is trying to
make this difficult. The program in Listing 3 is correct
with respect to the assertion that is expressed (e.g. z ==
30) as is evident by simple inspection of the code. The
program snippet in Listing 4 is the “after” program where
a non-linear opaque predicate transformation (see Sec-
tion 3) has been applied to hide the fact that at program’s
end, the value of z is in fact 30. In this case, it follows
from the fact that ∀x ∈ Z, ((x2 + x) mod 2) = 0.

This paper describes steps towards implementing a
framework in the Coq Proof Assistant [2] for a simple
imperative language that allows us to study obfuscating
transformations, their impact on programs, and how spec-
ified behaviour may be preserved beyond the transforma-
tions. Our starting point is the IMP language from [18],
which includes an encoding in Coq of a familiar small im-
perative language along with its formalized semantics. A
number of initial goals and principles drove the direction
of this research:

1) We don’t want to reinvent the wheel, which is why
we start with IMP.

2) We want to assure accessibility to as wide an audi-
ence as possible. For this reason, we choose IMP over
CompCert and Clight, which are used in [5]. On the
one hand, building on CompCert would have given
us lots of proofs and formalisms for free; however,
the significant learning curve associated with this
infrastructure seemed prohibitive, and thus much less
accessible.

3) We want the framework to be extendable. Following
the lead of [18], where a number of extensions to
IMP are easily added and studied, we wanted the
ability to build our obfuscation infrastructure incre-
mentally on top of IMP.

int main (int argc, char *argv[])

{

unsigned int x = 10;

unsigned int y = 20;

unsigned int z = 0;

z = x + y;

assert(z == 30);

return 0;

}

Listing 3: Original Code

int main (int argc, char *argv[])

{

unsigned int x = 10;

unsigned int y = 20;

unsigned int z = 0;

unsigned int a = ((unsigned int)argc);
unsigned int w = a * a;

w = a + w;

w = w % 2;

if (w == 0)

{

z = x + y;

}

else
{

z = y - x;

}

assert(z == 30);

return 0;

}

Listing 4: Obfuscated Code

Keeping these research goals in mind the contributions
of this paper are the following:
• We consider different formulations of what it means

for a transformation to be semantics-preserving: we
cover command equivalence in Sections 2.2 and 3.1
as well as Hoare logic equivalence in Section 3.2.
By command equivalence, we mean that if the two
programs start out in the same state, then the state
resulting after execution (just before termination, in-
termediate states don’t matter) of each program is
exactly the same. By state, we mean variable names
and their values; these values can change during ex-
ecution, but at the end, all values must have changed
in the same way. Hoare logic equivalence is a weaker
notion that is enough for most code obfuscation
transformations. It allows an obfuscated program to
use and modify variables that don’t occur in the
assertion. Only the values of variables that occur in

the assertion are required to have the same values at
the end of execution. Using Hoare logic equivalence
in this particular setting is a novel approach, helping
us establish a central strategy to “Formal Verification
of Program Obfuscation”: our obfuscated programs
will be “decorated” à la Pierce [18] with additional
assertions whose proofs will also be provided. The
additional assertions that our obfuscators will provide
will be needed to show the original desired behaviour
still holds after an obfuscation.

• We give motivations and a top-level explanations for
all of the transformations and one of the proofs (the
main one in Section 4). Proofs in Coq are done by
giving step-by-step commands called tactics, which
break the goal (the statement to be proven) into
subgoals until each branch of the proof is completed.
(We don’t discuss tactics, but refer the reader to [14]
for a more detailed treatment of both the proofs and
tactics.) To the best of our knowledge, the existing
literature does not provide this level of detail. We
provide not only an accessible explanation of ob-
fuscation techniques but also their formalization and
verification inside Coq. This follows along the lines
of research goal 2.

• We begin with a minimal version of IMP and aug-
ment it as needed for control flow flattening algo-
rithms (see Section 4), first by augmenting its syntax
and semantics with switch statements, and then by
defining a lower-level block-structured language that
wraps around blocks of code in order to model real-
world intermediate languages used in obfuscation
tools. This contributes to research goals 2 and 3.

This project, and in particular, the formalization in
Coq, was motivated by the Cloakware obfuscation tool
produced at Irdeto, and the interest in showing the cor-
rectness of some of its core functionality in safety critical
environments.1 For details of the Coq code and fuller
explanations of the proofs, the reader is referred to [13]
and [14], respectively.

2. The IMP Language and Hoare Logic

We now give the necessary definitions from Software
Foundations [18] needed to understand our formalization
in the sections that follow. Along the way, we describe
the features of Coq that we use. For more details on IMP,
the reader is referred to [18].

2.1. Syntax and Semantics of IMP

Informally, IMP commands are defined by the
following BNF grammar: c ::= SKIP | x ::=
a | c ;; c | IFB b THEN c ELSE c FI |
WHILE b DO c END. In [18], this grammar is encoded
by defining a new type called com and new keywords
(called constructors in Coq), one for each element of
the grammar, and then defining notation that maps
this formal definition to the BNF syntax. For example,

1. Much of it was carried out during the first author’s co-op term at
Irdeto under the supervision of the second author. The first author further
developed it in his Master’s thesis [14]; all lemmas and theorems referred
to in this paper are formally proved in Coq.

the definition of com includes the following case for
assignment statements: CAss (x : string) (a :
aexp), which introduces the CAss constructor and
specifies its two arguments along with their types. A
variable is represented using the built-in string type
in Coq, the arithmetic expression on the right of an
assignment command has type aexp, and the notation
specifies that Coq terms of the form (CAss x a) will
be written x := a. The definition of aexp in [18] is
defined using the same technique—starting from a BNF
grammar, constructors are defined and then notation
is given so that the syntax matches the grammar. We
omit its definition as well as the definition of bexp for
representing boolean expressions, which appear in IFB
and WHILE commands.

To define the operational semantics of IMP, the notion
of a state is needed. We introduce a type called state
defined to be an abbreviation for the type string ->
nat. Here, nat is Coq’s built in type for natural numbers
N. Although, we restrict our formal development to natural
numbers, it could be easily extended to other types such
as int or float. An initial state in Coq is a function
that maps all variables to the default value 0. States
are modified using an update operator. For example, the
state s & { X --> 3; Y --> 1} represents a state
in which s is modified so that X has the value 3 and Y
has the value 1. The “s &” can be omitted if the update
is to an initial state.

Evaluation of commands can be viewed as a partial
function from a command and a starting state to a new
state. It is encoded as a relation in Coq, and the notation
c / s1 \\ s2 means that starting in state s1 and
executing c results in a new state s2. We say that “c takes
s1 to s2.” The definition of this relation relies on two
functions. The first is aeval; the expression (aeval s
a) evaluates a (an element of aexp) in a state s, resulting
in a natural number. The function beval is similar and
maps a bexp to a Coq boolean. (We omit the details of
these definitions.)

2.2. Command Equivalence Definition

For two commands (IMP programs) c1 and c2 to be
command equivalent means that for any pair of states st
and st’, c1 takes st to st’ if and only if c2 takes st
to st’. In Coq, cequiv [18] is defined as a predicate
that takes two commands as arguments:
Definition 2.1 (Comand equivalence).
Definition cequiv (c1 c2 : com) : Prop
:= forall (st st' : state),

(c1 / st \\ st') <-> (c2 / st \\ st').

Prop is the type of logical formulas in Coq, and forall
and <-> are the notations for universal quantification
and logical equivalence, respectively. The corresponding
definitions for arithmetic expressions and booleans are
also important.
Definition 2.2 (Arithmetic/boolean equivalence).
Definition aequiv (a1 a2 : aexp) : Prop
:= forall s, aeval s a1 = aeval s a2.

Definition bequiv (b1 b2 : bexp) : Prop
:= forall s, beval s b1 = beval s b2.

2.3. Hoare Logic

Hoare logic is a way for us to prove that executing a
program will result in satisfying certain post-conditions,
(possibly) conditional on certain pre-conditions being met.
This involves defining a natural way of writing program
specifications, along with a compositional proof technique
to prove correctness with respect to them.
Definition 2.3 (Hoare triple). A Hoare triple is a triple

consisting of a pre-condition P , a program c, and a
post-condition Q, written

(|P |) c (|Q|),

which specifies that whenever P is true before execu-
tion, running the program c is guaranteed to make Q
true after execution.

This informal definition leaves states implicit, but for the
formal definition in Coq we will need to take states into
account.
Definition 2.4 (Assertion). An assertion about a program’s

state is a function from states to propositions.
Definition Assertion := state -> Prop.

Informally, for some assertion P and some state st, the
proposition (P st) represents the statement that P holds
in state st. As an example, let st be the state where the
value of every variable is 0. Let P be the assertion that
x = 0. Then (P st) is the proposition “x = 0 in the
state st”. In Coq, Hoare triples have three arguments, a
command and two assertions:

Definition hoare_triple (P:Assertion)
(c:com) (Q:Assertion) : Prop :=
forall st st', c / st \\ st' ->
P st -> Q st'.

and expresses that if P holds in state st and c takes st
to st’, then Q holds in st’. We will sometimes use the
notation {{P}}c{{Q}} to represent the Coq expression
(hoare_triple P c Q).

Proving that a Hoare triple holds is a line-by-line
affair, starting from the bottom of a program and working
upwards. There is one rule for each kind of IMP com-
mand, and the application is mostly mechanical, except
for the WHILE statement (see [18]).

3. Opaque Predicates in IMP/Coq

An opaque predicate [16] is a predicate that always
evaluates to either true or false and the truth-value is
known to the transformation but hard to deduce by an
attacker [7]. It could be any boolean expression in a pro-
gram, but we will only be concerned with those involving
arithmetic formulas here. An opaque predicate is used,
for example, in an IFB statement, and in the case when
it evaluates to true, the code under the false branch is
never evaluated at runtime. Thus opaque predicates incur
no runtime performance penalty.

Of course, the most basic opaque predicates are just
the boolean constants true and false themselves, but these
are not very useful in practice because it is immediately
obvious what is happening in the program, and neither
the simplest of humans nor tools will be fooled. For a

more advanced treatment of opaque predicates and how
they may be broken see [7].

An opaque predicate transformation takes as inputs
a program to be obfuscated, c1, an opaque predicate P
that takes one argument, and a dummy program c2, and
returns the program:

IFB (P x) THEN c1 ELSE c2 FI.

It is not known to an attacker, a priori, that c2 is a dummy
program. In practice, c2 should be complicated enough to
appear that it could feasibly be intended to be executed.
For example, c2 is often an actual command/statement
randomly chosen to be the dummy program.

In Section 3.1, we describe our initial (straightforward,
naive) attempt, in which the transformation introduces
new variables (as is standard in writing code in a typical
imperative language), assigns them values and then uses
them in the opaque predicate. Using this approach, trying
to state a general theorem about command equivalence
ends up being problematic.

We then discuss how this problem spawned two ideas
in different directions, which both rectify the issue. First,
we keep the above approach to adding variables, values,
and an opaque predicate, but use Hoare logic to prove as-
sertions that are weaker than command equivalence (Sec-
tion 3.2). Second, we reformulate the transformation to
rely on values already existing in the state of the program.
Here, we assume that the opaque predicate transformation
is being applied to a small piece of code somewhere within
a much larger program, one where these values already
exist in the state (Section 3.3).

3.1. Command Equivalence

Definition 3.1 (Factorial program, countdown nonzero
formulation). The following IMP program computes
the factorial of a nonzero natural number. The input
is read from X, temporary values are stored as Z, and
the factorial of the input is stored in Y as the output.

Definition fact_nonzero : com :=
Z ::= X;;
Y ::= 1;;
WHILE ! (Z <= 1) DO
Y ::= Y * Z;;
Z ::= Z - 1

END.

The choice of factorial program as a candidate for exam-
ples of obfuscation is somewhat arbitrary. It works well
for illustrative purposes, as it is neither too complex nor
completely trivial.
Example 3.2. The fact_nonzero program with input

X = 3 yields output Y = 6.
Example factorial_3:
fact_nonzero / { X --> 3 } \\
{ X --> 3; Z --> 3; Y --> 1; Y --> 3;
Z --> 2; Y --> 6; Z --> 1 }.

The Example keyword in Coq is followed by a
name, a colon, and a statement (a logical proposition
in Coq), which is then followed by a proof. In this
case, the statement expresses that the fact_nonzero
program takes the state { X --> 3 } to the state

{ X --> 3; Z --> 3; Y --> 1; Y --> 3;
Z --> 2; Y --> 6; Z --> 1 }. Note that the
final state contains the information for every intermediate
assignment made by the program. The rightmost value
of each variable contains its final value and thus we can
discern the correct output Y = 6. (For the Coq proof of
this statement, and all other examples and theorems in
this and the next section, refer to [13].)

For this section, we’ll use the following fact to define
a simple opaque predicate:

∀x. (x ∗ x+ x+ x+ 1) = (x+ 1) ∗ (x+ 1).

We now define an opaque predicate transformation with
our running example. For the purposes of making the
proofs easier to work with, and also to add a slight addi-
tional touch of obfuscation, we split up these assignments
over multiple lines, as follows.

Definition opaque_trans x c1 c2 :=
X' ::= (ANum x) ;;
Z' ::= X' * X' ;;
Z' ::= Z' + X' ;;
Z' ::= Z' + X' ;;
Z' ::= Z' + 1 ;;
Z'' ::= X' + 1 ;;
Z'' ::= Z'' * Z'' ;;
IFB (BEq Z' Z'') THEN c1 ELSE c2 FI.

Here (ANum x) is the representation of the number
x as an arithmetic expression (of type aexp). The
opaque_trans function takes three arguments as input:
a number x, and programs c1 and c2, and returns the new
program that executes c1 if the equation

(x ∗ x+ x+ x+ 1) = (x+ 1) ∗ (x+ 1)

holds and executes c2 otherwise. Of course, the above
equation is true for all x (which we have proved in
Coq), so the resulting program should be the same as
c1. We’d like to claim that a program transformed by
opaque_trans is equivalent to the original.

What do we mean when we say the transformed
program should be “equivalent”? Example 3.3 shows
what happens when opaque_trans is applied to the
fact_nonzero example.
Example 3.3. For any x ∈ N and any program c2,

opaque_trans x fact_nonzero c2 with in-
put X = 3 yields output Y = 6. In Coq, however,
it looks as follows.

Example factorial_3_opaque_trans:
forall x c2,
opaque_trans x fact_nonzero c2 /
{ X --> 3 } \\
{ X --> 3; X' --> x; Z' --> x * x;
Z' --> x * x + x;
Z' --> x * x + x + x;
Z' --> x * x + x + x + 1;
Z'' --> x + 1;
Z'' --> (x + 1) * (x + 1);
Z --> 3; Y --> 1; Y --> 3;
Z --> 2; Y --> 6; Z --> 1 }.

The proof of this statement begins (see [13]) by intro-
ducing the quantified variables and unfolding defini-
tions; after which the transformed program becomes:

X' ::= x;;
Z' ::= X' * X';;
Z' ::= Z' + X';;
Z' ::= Z' + X';;
Z' ::= Z' + 1;;
Z'' ::= X' + 1;;
Z'' ::= Z'' * Z'';;
IFB Z' = Z''
THEN Z ::= X;; Y ::= 1;;

WHILE ! (Z <= 1) DO
Y ::= Y * Z;; Z ::= Z - 1

END
ELSE c2 FI

Having a proof of this theorem illustrates that our
opaque predicate transformation worked; however, there
seems to be no direct way to generalize Example 3.3 in
terms of command equivalence. We cannot use cequiv
(Definition 2.1)—that is, we can’t use it with the current
formulation of the transformation—since the transforma-
tion introduces new variables, which affect the value of
the end state, even if those variables are not of interest
to us. In particular, note that although variable Y has the
same value in the end states in Examples 3.2 and 3.3, the
rest of the state information is not identical. We’ll revisit
command equivalence in Section 3.3.

3.2. Hoare Logic Equivalence

In this section, we explore using Hoare logic to specify
program conditions, and then generalize the result as much
as we can. The main idea with Hoare logic is that we can
be more specific about what we wish a transformation to
preserve which it turns out is often good enough. First,
we’ll use a slightly different formulation of the factorial
program.
Definition 3.4 (Factorial program, count-up formula-

tion). This version of the factorial program counts up
from zero rather than down from X, and works for
input 0 as well.

Definition fact_program : com :=
Y ::= 1;;
Z ::= 0;;
WHILE ! (Z = X) DO
Z ::= Z + 1;;
Y ::= Y * Z

END.

We begin by defining assertions about the values of X
and Y in Coq as follows.

Definition as_x (x0 : nat) : Assertion
:= (fun st => st X = x0).

Definition as_y (y0 : nat) : Assertion
:= (fun st => st Y = y0).

We restate Example 3.2, replacing the specific values of
3 and 6 with arbitrary natural numbers, expressed as the
following Hoare triple:
Example 3.5. (|X = x0|) fact program (|Y = x0!|),

which in Coq is:
Example factorial_all_hoare: forall xo,

{{ as_x xo }}
fact_program

{{ as_y (fact xo) }}.

Example 3.5 states that when X = x0 before the (unob-
fuscated) fact_program runs, then Y = (fact x0)
afterward, where fact is the mathematical definition of
the factorial function expressed in Coq.

We now turn to showing that when we obfuscate
fact_program, it remains the case that X = x0 be-
forehand implies that Y = (fact x0) when the pro-
gram finishes. In the following, we use a new formulation
of the opaque predicate transformation, where we collapse
the assignments into single lines.

Definition opaque_trans' x c1 c2 :=
X' ::= (ANum x) ;;
Z' ::= X' * X' + X' + X' + 1 ;;
Z'' ::= (X' + 1) * (X' + 1) ;;
IFB (BEq Z' Z'') THEN c1 ELSE c2 FI.

Example 3.6. We now apply the transformation
opaque_trans’ to fact_program to obtain the
following statement expressing that the same Hoare
triple holds with the obfuscated factorial program in
place of the original program.

∀x, x0, c2
(|X = x0|) opaque trans′ x fact program c2
(|Y = x0!|)

which in Coq is:
Example factorial_all_hoare_opaque:
forall x xo c2,
{{ as_x xo }}

(opaque_trans' x fact_program c2)
{{ as_y (fact xo) }}.

At this point, we’ve stated that our factorial program
satisfies a Hoare triple of the form:

(|X = x0|) c (|Y = y0|).

where c is factorial or the transformed version of factorial.
Indeed, the decision to use the factorial program in the
previous examples for illustrative purposes was an unnec-
essary detail, so we show how to generalize to an arbitrary
program by introducing a new term Hoare fidelity, and
then proving a general theorem.
Definition 3.7 (Hoare fidelity, with respect to input X and

output Y). A program c2 preserves the Hoare fidelity
of a program c1 with respect to input X and output Y,
if the validity of the Hoare triple

(|X = x0|) c1 (|Y = y0|)

implies the validity of the Hoare triple

(|X = x0|) c2 (|Y = y0|).

In Coq,
Definition Hoare_fidelity_xy c1 c2
:= forall xo yo,
hoare_triple (as_x xo) c1 (as_y yo) ->
hoare_triple (as_x xo) c2 (as_y yo).

Theorem 3.8. For all programs c1 and c2, and all x ∈ N,
the transformed program (opaque_trans’ x c1
c2) preserves the Hoare fidelity of c1 with respect
to input X and output Y. In Coq,

Theorem Opaque_trans_hoare_fidelity_xy:
forall x c1 c2,
Hoare_fidelity_xy c1
(opaque_trans' x c1 c2).

As a result, the theorem factorial_all_hoare_-
opaque in Example 3.6 now follows directly from this
general theorem and theorem factorial_all_hoare
in Example 3.5 about the original program before trans-
formation.

3.3. A Formulation without Assignment

In the first presentation of the opaque predicate trans-
formation from Section 3.1, we used a program that
allowed the user (that is, the person obfuscating the code)
to specify a particular number, and then add a number of
assignments before the opaque predicate check, and then
ultimately noted at the end of Section 3.1 that command
equivalence (which depends on the full state — that is,
the equality of values of all variables) did not hold in this
model due to these extra assignments and variables.

We now present an alternate formulation with no as-
signments, with the entire predicate built into the boolean
condition of the branching statement. On the one hand,
the entire equation appears on a single line instead of in
a number of assignments, which may make it easier to
detect, but on the other hand, it can access any variable
already being used (and in the case of IMP, any variable
at all; recall a state in IMP is a total map from strings
to N and all variables have default value 0). In this case,
command equivalence can be proven in general.

The following two definitions are important in the
theorems and examples below.

Definition
make_opaque_pred (a1 a2: aexp): bexp

:= BEq a1 a2.
Definition

make_opaque_pred_IFB b c1 c2
:= IFB b THEN c1 ELSE c2 FI.

Theorem 3.9. If a boolean b is equivalent to true, then
for any programs c1 and c2, the program c1 is
command equivalent to the program resulting from
applying make_opaque_pred to b, c1, and c2.

The power of the result stated in Theorem 3.9 is
that now, the particular programs and predicate used are
irrelevant and can be swapped with anything, so long as
we can prove the fact that the boolean supplied is indeed
an opaque predicate.

Example 3.10. We can now apply this general theorem
to our running example of predicate and factorial
program.

Example example_fact_opaque_pred:
cequiv fact_nonzero
(make_opaque_pred_IFB
(make_opaque_pred
((X + 1) * (X + 1))
(X * X + X + X + 1))
fact_nonzero SKIP).

4. Control Flow Flattening in IMP/Coq

Today’s reverse engineering tools and/or other static
analysis tools can at a glance reveal some information
about the control flow of the program, or the rough struc-
ture as delineated by the flow of blocks of code through If-
Then-Else, While-Do-End, Switch, and Jump constructs.

One of the obfuscation techniques to make this diffi-
cult to analyze is control flow flattening. This technique
aims to break apart all of the constructs that would reveal
information about a program’s control flow and flatten an
entire program into a single semantically equivalent switch
statement inside a while loop. In this section (subsections
4.1 and 4.2), we focus on a single transformation that
turns an If-Then-Else construct of the form in Figure 1 to

Figure 1. If-Then-Else

the equivalent flattened program illustrated in Figure 2.

Figure 2. Flattened If-Then-Else

In Section 4.3, we briefly discuss other transformations
that we have defined and proven correct but do not have
space to provide details for. For a thorough treatment
of control flow flattening obfuscation and its effects in
obstructing static analysis see [19].

4.1. Augmenting IMP with Switch (IMP+Switch)

Before we can formalize control flow flattening of an
If-Then-Else construct, we need to enrich IMP with the
syntax and semantics of switch statements. We extend the
original IMP language from [18] and add the clause |
SWITCH v l to the BNF grammar presented in Sec-
tion 2.1. Here v is the name of the switch variable (which

is s in Figure 2), represented as a Coq string. The
second argument l is a list of pairs (represented using
Coq’s built-in types for lists and pairs), where the first
element of each pair is in N, representing the case number
in the switch statement, and the second element is a com
representing the program associated with that number. We
accordingly redefine the command evaluation semantics
(the definition of c / s1 \\ s2) to include evaluation
of switch statements. In addition, we extend the proof
in [18] to show that evaluation is still deterministic with
our new switch statements added to IMP.

4.2. Flattening If-Then-Else in IMP+Switch

We wish to prove command equivalence between the
original and transformed programs. We note that the
switch variable s is introduced in Figure 2 with value
0 and ends with value 5. Hence, we need to preprocess
the original program to be transformed by adding in
these assignments. We define two functions in Coq, one
for preprocessing and one to transform the If-Then-Else
statement into a switch statement. Both functions take
a boolean condition, and two subprograms for the if-
then and if-else branches. They also take a header and
footer, which are subprograms that are executed before
and after the If-Then-Else statement, respectively. These
functions appear in Listings 5 and 6. The difference

Definition preprocess_program

header cond c1 c2 footer : com :=

swVar ::= 0 ;;

header ;;

IFB cond THEN c1 ELSE c2 FI ;;

footer ;;

swVar ::= 5.

Listing 5: The preprocess_program function

Definition transform_program

header cond c1 c2 footer : com :=

swVar ::= 0 ;;

WHILE (swVar <= 4) DO

SWITCH swVar [

(0, header ;; swVar ::= 1) ;

(1, IFB cond THEN swVar ::= 2

ELSE swVar ::= 3 FI) ;

(2, c1 ;; swVar ::= 4) ;

(3, c2 ;; swVar ::= 4) ;

(4, footer ;; swVar ::= 5) ]

END.

Listing 6: The transform_program function

is that the first function forms a regular If-Then-Else
statement, and the second one forms the flattened version.
From the components of the original program, it builds a
switch statement wrapped in a while loop, with the cases
appropriately handled.

Before we state the general theorem that any pro-
gram’s preprocessed and transformed forms are command
equivalent, we have to fully account for the newly intro-
duced switch variable which controls the switch statement,

called swVar in Listings 5 and 6. If the original program
already uses this variable in some way, then command
equivalence will not necessarily hold. For example, sup-
pose the header of the program to be transformed contains
the assignment swVar := 999. This would then com-
pletely bypass the entire flattened switch construct! We
avoid this problem with the definition below of evaluation
invariance, followed by a lemma about it.
Definition 4.1 (Evaluation invariance). A program c is

evaluation invariant with respect to a variable X if, for
all states st and st’ and all n ∈ N, c evaluates st
to st’ if and only if c evaluates st updated with X
--> n to st’ updated with X --> n. In Coq:

Definition eval_invariant c X :=
forall n st st', c / st \\ st' <->
c / st & { X --> n }

\\ st' & { X --> n }.

In other words, if the only thing that changes about
the start state is the value of X, there is no impact on
evaluation with the sole exception of the same change to
X in the end state.
Lemma 4.2. Evaluation invariance implies evaluation in-

dependence in the sense that, if a command c is eval-
uation invariant with respect to X, then if c evaluates
a state st updated with X --> n for some n ∈ N to
st’, then c also evaluates st to st’. In Coq,

Lemma eval_inv_imp_eval_ind:
forall c X n st st',
eval_invariant c X ->
c / st & { X --> n } \\ st' ->
c / st \\ st'.

In order to state and prove the main theorem, we also
need the related concept of boolean invariance.
Definition 4.3 (Boolean invariance). A boolean expres-

sion b is boolean invariant with respect to a variable
X if for all states st and all n ∈ N, the boolean
evaluation of b in st is the same as the boolean
evaluation of b in st updated with X --> n. In Coq:

Definition beval_invariant b X :=
forall n st, beval st b =
beval (st & { X --> n }) b.

We can now state the main theorem. In particular,
control flow flattening of If-Then-Else constructs is sound
in the following sense:
Theorem 4.4. Fix the variable swVar for the control flow

flattening transformation. For any program of the form
header;;IFB cond THEN c1 ELSE c2 END
;; footer, we have command equivalence between
the following two programs (which we will call p1
and p2 in the proof below):
preprocess_program header cond

c1 c2 footer
transform_program header cond

c1 c2 footer

as long as the following hold:
• The commands footer, c1, and c2 are evaluation

invariant with respect to swVar.
• The boolean condition cond is boolean invariant

with respect to swVar.

cond : bexp
header, c1, c2, footer : com
st, st', s0, s1, s2, s3 : state
HI1 : eval_invariant c1 swVar
HI2 : eval_invariant c2 swVar
HIf : eval_invariant footer swVar
HB : beval_invariant cond swVar
H2 : (swVar ::= 0) / st \\ s0
H3 : header / s0 \\ s1
H4 :
(IFB cond THEN c1 ELSE c2 FI)
/ s1 \\ s2

H6 : footer / s2 \\ s3
H11 : (swVar ::= 5) / s3 \\ st'
A0 : aeval s0 swVar = 0
A5 : aeval st' swVar = 5
Hcond : beval s1 cond = true
============================
(swVar ::= 0;;
WHILE swVar <= 4
DO SWITCH swVar

[(0, header;; swVar ::= 1);
(1, IFB cond THEN swVar ::= 2

ELSE swVar ::= 3 FI);
(2, c1;; swVar ::= 4);
(3, c2;; swVar ::= 4);
(4, footer;; swVar ::= 5)]

END) / st \\ st'

Figure 3. Command equivalence proof step

In Coq:
Theorem AllTransEquiv:
forall header cond c1 c2 footer,
eval_invariant c1 swVar ->
eval_invariant c2 swVar ->
eval_invariant footer swVar ->
beval_invariant cond swVar ->
cequiv (preprocess_program header

cond c1 c2 footer)
(transform_program header

cond c1 c2 footer).

Proof Sketch. Proving command equivalence in this
case requires proving that for all states st and st’,
both p1 / st \\ st' -> p2 / st \\ st' and
p2 / st \\ st' -> p1 / st \\ st'. Each
case has two subcases depending on whether cond is
true or false. We discuss only the first case, with the
subcase where cond is true. In the Coq proof, we work
in the backward direction, starting with the statement
of the theorem and breaking it down into subgoals
step-by-step by applying tactics. After numerous steps,
the subcase of interest is expressed as the Coq goal in
Figure 3.

The formulas above the double line are the current
hypotheses, while the statement to be proven is below the
line. Many of the hypotheses come from breaking down
the program p1 into evaluation steps for single statements.
They come from the definition of the evaluation relation,
which as discussed, defines the operational semantics of
IMP+Switch. For example, H2 expresses that the assign-
ment statement swVar ::= 0 is evaluated in the start

state st resulting in the intermediate state s0, while
H3 expresses that evaluating the header starting from the
intermediate state s0 results in a new intermediate state
s1. The last 3 hypotheses state some facts about the values
of swVar in the start and end states and show the value
of cond for this subcase. To prove the conclusion, the
definition of evaluation is again important, along with the
fact that the evaluation relation is deterministic.

4.3. Transforming a While-Do-End Construct

We now switch gears and briefly discuss dismantling
and then flattening a While-Do-End construct, focusing
on an example described in [19]. For this particular trans-
formation, we start with a program with a While-Do-
End construct, and dismantle it into a number of basic
blocks (where each block is a sequence of non-control
flow commands ending with a control flow command),
essentially replacing the While-Do-End construct with
conditional GoTo statements at the end of some blocks.
During execution, the targets of these GoTos are deter-
mined dynamically with conditions on some variable in
memory, instead of a direct (constant) address as the jump
target. Following [19] we call this intermediate trans-
formation a dismantling. Dismantling is then followed
by a flattening operation, reminiscent of the flattening
transformation above.

For this transformation, we modify IMP once again.
This time, we replace the switch command of the previ-
ous section with a lower-level language (which we call
IMP+Flow since it explicitly deals with flow of control
constructs like jumps) containing basic blocks, with a
switch statement as one of the kinds of blocks. This
language is similar to intermediate languages into which
higher-level programs are transpiled, and which are used
in commercial obfuscation tools such as Cloakware’s ob-
fuscation engine [12].

In Coq, we introduce a new type called comBlock,
which encodes the following grammar: bJump c l |
bCond c b l l | bSwitch c v m | bEnd c.
Each kind of block consists of a command together with
an instruction of how to find the next block. They include
1) an unconditional jump where the next block is denoted
by a label l represented as a Coq string, 2) a conditional
jump, where b is a bexp and the two labels indicate
where to jump in the true and false cases, 3) a switch
command, with switch variable v and mapping m, which
maps values of the switch variable to labels, and 4) a
terminal block. A program is then a pair consisting of
an initial command block and a block dictionary, which
maps labels to command blocks. We must also extend the
definition of the evaluation relation to command blocks.
(We omit the details. As usual, see [13], [14].)

We illustrate this notion of dismantling and flattening
with the example mentioned earlier. The original program
is a single block containing the IMP program below:

WHILE (A <= 2) DO
B ::= A + B ;;
IFB (!(B <= 4)) THEN
B ::= B - 1

ELSE SKIP FI ;;
A ::= A + 1

END ;;
RETURN ::= A * B.

We do not describe the transformation any further, but just
note that the transformed program contains the following
blocks, and execution starts at block L1.

"L1" --> bConditional SKIP
(! (A <= 2)) "L4" "L3";

"L2" --> bJump (A ::= A + 1) "L1";
"L3" --> bConditional (B ::= A + B)

(B <= 4) "L2" "L5";
"L4" --> bEnd (RETURN ::= A * B);
"L5" --> bJump (B ::= B - 1) "L2"

We have proved that both versions of the program, when
starting in a state with A = 1 and B = 2, evaluate to
the same final state with RETURN = 12. Future work
includes extending our command equivalence theorem to
this kind of transformation.

5. Obfuscation in Coq: Related Work

There have been three papers by Sandrine Blazy and
co-authors that study code obfuscation in Coq, which we
discuss here.

The first paper [3] also uses IMP (their own formu-
lation and not the one from [18]) as the language for
obfuscation, but studies data and layout obfuscation tech-
niques, as opposed to the control obfuscation techniques
which opaque predicates and control flow flattening fall
under [6]. They first consider obfuscating integer constants
using a function that maps each integer to a different one,
and then performing a substitution using this mapping.
They also consider changing variable names, which can
be used, for example, to change a descriptive variable
name like account balance to a string of gibberish. Such
techniques are inherently different from the ones we have
studied. One can make a simple combinatorial argument
that putting them together in the same obfuscation trans-
formation would generate a synergistic effect, making a
program possibly much more difficult to analyze.

The second paper [5] also studies control flow flatten-
ing, but the authors use the Clight language of CompCert
[11] (the formally verified C compiler in Coq) and prove
the correctness of the obfuscation in that setting, from
which the correctness of the compilation process fol-
lows “for free” from CompCert’s own proofs of semantic
preservation.

On the one hand, the approach in [5] is less elementary
and less accessible, as it works with a nontrivial subset of
the real C language, but on the other it is clear evidence
that formal verification of obfuscation techniques need not
be restricted to a small language like IMP. Other real-
world practicalities considered in [5] include simulation
techniques and analysis of running time.

The work in [5] also discusses some solutions for
improving the stealth of control flow flattening, such as
splitting a switching variable into two different variables
that are updated at different points of a program, as well
as randomly encoding the values of the switch cases so
they don’t just correspond to consecutive numbers.

In comparing the work in [5] to ours, we believe there
is merit both in the IMP and the CompCert routes. In

the former, the language used is of minimal complexity,
which allows not only for specifications and proofs of
transformations to be developed quicker without being
bogged down in unnecessarily complicated features of
the underlying language, but is also better suited for
pedagogical purposes (see our research goal 2). IMP is
also Turing complete, so from a theoretical point of view
there is no loss of generality in proofs made using it—
they can always be adapted to CompCert later. On the
other hand, CompCert is, of course, closer to languages
that would be of interest to real-world applications and so
more practical in that sense.

The authors of [5] also needed to separate switching
variables from those in the program to be transformed
(as we did), but their solution was different. They instead
use a function to parse the program to be transformed
and generate a fresh variable which doesn’t appear there
to be used for the transformation. From a practical point
of view, this approach is perhaps more natural, and in
line with how a real obfuscating tool would function—
generating new variables rather than demand that a certain
specifically named variable doesn’t exist in the source
program. Theoretically, though, these are equivalent, since
any program can contain only finitely many variable
names, and there are an infinite number to choose from.

The third paper [4] continues to work in Clight, and
studies obfuscations that involve mixing arithmetic oper-
ators and bitwise boolean operators. This is another data
obfuscation that appears frequently in real-world binary
code, but as it is based on features wildly beyond the
capabilities of IMP, a detailed discussion is beyond the
scope of the present work.

6. Future Work

The work done to date on formal verification of ob-
fuscation, both in the present work and in the papers
of Blazy et al., while providing a solid proof-of-concept
that obfuscation tools of the future could support formal
verification, are still limited in scope in the sense that they
treat individual transformations.

A real world obfuscator mixes many different transfor-
mations together at once, often in non-deterministic ways
for diversification of obfuscations, and so some form of
compositionality would need to be implemented on these
formal proofs to be able to use them together and preserve
the desired formulation of correctness.

Earlier (first contribution item) we mentioned how our
obfuscators would generate additional assertions to show
the obfuscated program preserves a desired behaviour of
the original program. While we have worked out offline
proofs for certain obfuscations, such as some new opaque
predicates, we still need a way to integrate these proofs
into the machinery of “decorated programs” that IMP
already provides.

Furthermore, we (along with the work of Blazy et
al.) have, in the formal setting of Coq, only tackled
one desired property of obfuscation—correctness. That is,
some form of the semantics of the program, or relationship
between inputs and outputs, should be preserved (obfusca-
tion property 1, i.e., the functionality property). But there
are, of course, other properties that have not been touched
upon, namely properties 2 and 3.

In closing, we stress, once more, that it is important
to actually apply formal specifications and methods to
security goals and metrics in some form, so we can come
full circle and give prospective clients of an obfuscation
tool a clear answer to the other big question “How exactly
will using this improve the security of my programs?” and
be able to back our answer with a proof that it actually
does so.

